
1 Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 12.4.3.2a which is part of the UTRAN test suite. Only essential changes to the TTCN are applied and documented in section 4.

With these changes applied the test case can be demonstrated to run with one or more 3G UEs 
(see section 6). Execution log files are provided as evidence. 

3 Verification Test Summary

Test Case:
TC_12_4_3_2a
Test Group:
/12/UTRAN34_GMM_NIMTC.ttcn
ATS Version:
iwd-TTCN3-B2013-03_D15wk25 + essential modifications.

System Simulator used:
Rohde & Schwarz CMW500

UE used:
Intel 7260
Verification Status:
PASS
4 Corrections required for test case 12.4.3.2a
4.1 Introduction

This section describes the changes required to make test case 12.4.3.2a run correctly with a 3G UE. All modifications are marked with label “WA#” in the TTCN comments column of the enclosed ATS [1].

The ATS version used as basis was UTRAN testsuite which is part of iwd-TTCN3-B2013-03_D15wk25 release. This testsuite provided by MCC160 contains UTRAN test cases.
4.2 f_TC_12_4_3_2a_UTRAN()
	Record Name
	f_TC_12_4_3_2a_UTRAN()

	Reason for change
	In the curent TTCN implementation, the Cell power is not set correctly for the serving cell. This needs to be set correctly.

Function f_UTRAN34_AttachReject  followed by f_UTRAN34_SwitchPower_Off is being used for preamble. The pre condition says that the UE should have P-TMSI-1 and RAI1. These functions do not achieve this. This needs to be corrected.

	Summary of change
	Called function f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationServingCell); after cell creation.

Created a new function f_UTRAN34_Preamble_NIMTC and called it followed by f_UTRAN34_SwitchPower_Off(utran34_CellA, U1_IDLE);

	Source of change
	UTRAN34_GMM_NIMTC.ttcn

	
	


Before:
	function f_TC_12_4_3_2a_UTRAN() runs on UTRAN_PTC
  {
    //Initialise UTRAN cells according to 34.108
    f_UTRAN34_Init();
    //Initial Conditions:
    /*One cell operating in network operation mode II (in case of UE operation mode A).
     */
    if (pc_SupportOpModeA and not px_SupportOpModeC) {
      f_UTRAN_CellInfo_SetNMO(utran34_CellA, tsc_NMO_II);
    }
    // Cell1 started switched on with default power level
    f_UTRAN34_SS_CreateCellDCH(utran34_CellA);
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_CellA);
    // Insert correct USIM
    f_UT_USIM_Insert(UT, "34.123-1 clause 12.4.3.2a");
    // Preamble rejecting (Attach Reject) - The UE has a valid IMSI. The UE has been registered in the CS domain.
    f_UTRAN34_AttachReject(utran34_CellA);

    f_UT_SwitchOffUE(UT, false);

<<SKIPPED SECTION>>


After:

	function f_TC_12_4_3_2a_UTRAN() runs on UTRAN_PTC
  {
    //Initialise UTRAN cells according to 34.108
    f_UTRAN34_Init();
    //Initial Conditions:
    /*One cell operating in network operation mode II (in case of UE operation mode A).
     */
    if (pc_SupportOpModeA and not px_SupportOpModeC) {
      f_UTRAN_CellInfo_SetNMO(utran34_CellA, tsc_NMO_II);
    }
    // Cell1 started switched on with default power level
    f_UTRAN34_SS_CreateCellDCH(utran34_CellA);
    f_UTRAN_SetCellPower(utran34_CellA, tsc_AttenuationServingCell);//WA#12_4_3_2a
    //Send default system information
    f_UTRAN34_SendDefSysInfo(utran34_CellA);
    // Insert correct USIM
    f_UT_USIM_Insert(UT, "34.123-1 clause 12.4.3.2a");
    // Preamble rejecting (Attach Reject) - The UE has a valid IMSI. The UE has been registered in the CS domain.
    //WA#12_4_3_2a f_UTRAN34_AttachReject(utran34_CellA); 
    //f_UT_SwitchOffUE(UT, false);
    f_UTRAN34_Preamble_NIMTC (utran34_CellA);//WA#12_4_3_2a
    /*
    f_UT_SwitchOnUE(UT, false);//WA#12_4_3_2a This is for Initial condition
    f_UTRAN34_GMMOnly_Attach(utran34_CellA, omit, delayTolerantAccess);//WA#12_4_3_2a This is for Initial condition*/
    f_UTRAN34_SwitchPower_Off(utran34_CellA, U1_IDLE);//WA#12_4_3_2a This is for Initial condition
    //Further initial conditions
<<SKIPPED section>>


4.3 F_UTRAN34_Preamble_NIMTC
	Record Name
	f_UTRAN34_Preamble_NIMTC

	Reason for change
	Please see change 4.3 above

	Summary of change
	New function

	Source of change
	UTRAN34_CommonFunctions.ttcn

	
	


Before:
	function f_UTRAN34_Preamble_NIMTC(UTRAN_CellId_Type p_CellId) runs on UTRAN_PTC
  { /* @sic R5s130347: local GuardTimer sic@ */
    timer t_GuardTimer := tsc_GuardTimePreamble;
    t_GuardTimer.start;
    //  Switch cell power on
    f_UTRAN_SetCellPower(p_CellId, tsc_AttenuationServingCell);
    //  Switch on UE
    f_UT_SwitchOnUE(UT, false);
    //  Establish RRC Connection, perform Authentication, Security Mode procedure,
    //  Attach procedure and release RRC Connection
    f_UTRAN34_IdleUpdated(p_CellId, omit, delayTolerantAccess);
    t_GuardTimer.stop;
  }


4.4 fl_TC_12_4_3_2a_Body()
	Record Name
	fl_TC_12_4_3_2a_Body()

	Reason for change
	In the current TTCN, according to the expected test sequence in prose, the UE is made to perform a ATTACH REQ procedure (Steps 4 to 8). For this, the Cell A needs to be configured correctly. This is not implemented and needs to be done.

The current TTCN expects the establishment cause in the rrc connection request at Step 3 to be “registration”. However, as per 24.008 Annex L, table L.1.2, the RRC Establishment cause for Any PS NAS procedure where the initiating uplink signalling message has the Device properties IE with low priority indicator set to "MS is configured for NAS signalling low priority" will be set as delayTolerantAccess. Thus, the establishment cause in the rrc connection request at Step 3 should be delayTolerantAccess, not registration. This needs to be changed in the current TTCN. This also affects Step 11.
In current TTCN, the variablev_MobileIDPTMSI_1 is used to check the mobile identity sent by the UE in the ATTACH REQ message at Step 4. This variable is being set by assignment v_MobileIdPTMSI_1 := cs_MobileIdTMSIorPTMSI('18'O, px_PTMSI_Def); which sets the IEI to ‘18’O. This is incorrect, the IEIshould be set to omit. This needs to be changed.

In the current TTCN, in the ATTACH ACCEPT message that is transmitted at Step 8, the P-TMSI and P-TMSI-signature is being set incorrectly and need to be corrected.

The ROUTING AREA UPDATE REQ at Step 12 is being expected as UpLinkDirectTransfer. This is incorrect since this is the first NAS message on the new rrc connection. This needs to be changed to InitialDirectTransfer.

Current TTCN expects the UE to transmit a RAU Complete message. However, this is incorrect since in the RAU ACCEPT message no new mobile identity is provided. Also, the Prose doesn’t expect the RAU complete. This needs to be removed.

	Summary of change
	Called the function f_UTRAN_PrepareCellRRC_ConnEst(utran34_CellA, utran34_CellA); before the ATTACH procedure on Cell A be performed.

Changed the establishment cause in the Step 3 and Step 11 to delayTolerantAccess.

Assigned variable cs_MobileIdTMSIorPTMSI(omit, px_PTMSI_Def);

Corrected the ATTACH ACCEPT message by sending right values for P-TMSI and P-TMSI-Signature.

Changed the template for capturing the RAU REQ from car_PS_UplinkDirectTransfer to car_PS_InitDirectTransfer

Removed the implementation to wait for RAU COMPLETE message (in blue)
Also assigned intermediate verdicts to improve the test flow traceability

Please see screenshot below.

	Source of change
	UTRAN34_GMM_NIMTC.ttcn

	
	


Before:
	function fl_TC_12_4_3_2a_Body() runs on UTRAN_PTC
  {
    var template (value) MobileIdentity v_MobileIdPTMSI_1;
    var RRC_DATA_IND v_RRC_DataInd;
    var U_RLC_TM_IND v_RLC_TR_DATA_IND; //@sic R5-150124 sic@
    var boolean v_NewKeyFlag := true;
    var START_Value v_START_Value_ps := '00000000000000000000'B;
    var PLMN_Identity v_PLMN_Identity;
    var octetstring v_PLMN;
    var octetstring v_LAC;
    var octetstring v_RAC;
    var template (present) RoutingAreaId v_RAI_1;
    var InitialUE_Identity v_InitialUE_Id;
    timer t_T3312Ext_UpperBound := 60.0 * 6.0 * 1.1; // 6 minutes + 10 % tolerance
    timer t_T3312Ext_LowerBound := 60.0 * 6.0 * 0.9; // 6 minutes - 10 % tolerance
    v_PLMN_Identity := f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellA);
    v_PLMN := f_U_Asn2Nas_PlmnId(v_PLMN_Identity);
    v_LAC := f_UTRAN_CellInfo_GetLAC(utran34_CellA);
    v_RAC := f_UTRAN_CellInfo_GetRAC(utran34_CellA);
    v_RAI_1 :=   cr_RAI(omit, cr_LAI(omit, v_PLMN, v_LAC), v_RAC);
    v_MobileIdPTMSI_1 := cs_MobileIdTMSIorPTMSI('18'O, px_PTMSI_Def);
    //@siclog "Step 2" siclog@
    //The UE is powered up or switched on and initiates an attach (see ICS).
    f_UT_SwitchOnUE(UT, false);
    // For non auto attach UE: trigger the Attach procedure via an AT command.
    if (not(pc_AutomaticAttachSwitchON)) {
      f_Delay(5.0);
      f_UT_TriggerAttach(UT);
    }
    //@siclog "Step 3" siclog@
    //SS checks that the IE "Establishment cause" in the received RRC CONNECTION REQUEST message is set to "Registration".
    f_UTRAN34_RRC_ConnEst(utran34_CellA, registration); 
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 3");
    //@siclog "Step 4" siclog@
    /*Attach type = 'GPRS attach'
     * Mobile identity = P-TMSI-1
     * Old Routing area identity = RAI-1
     * "MS network feature support": 1 (MS supports the extended periodic timer in this domain)
     */
    U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U34_AttachReq(cr_AttachType(?, tsc_I_AttachTypeGPRSAtt),
                                                                                          v_MobileIdPTMSI_1,
                                                                                          v_RAI_1,
                                                                                          ?)))
      -> value v_RRC_DataInd;
    if (v_RRC_DataInd.msg.attachRequest.msNetworkFeatureSupport.extdPeriodicTimers == '1'B) {
      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 4");
    }
    else {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4");
    }
    v_START_Value_ps := v_RRC_DataInd.ttcn_start;
    f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);
    //@siclog "Steps 5-6" siclog@
    f_UTRAN_GMM_Authentication();
    //@siclog "Step 7" siclog@
    f_UTRAN34_RRC_Security(utran34_CellA,     // SECURITY MODE COMMAND, SECURITY MODE COMPLETE
                           v_NewKeyFlag,
                           ps_domain);
    //@siclog "Steps 8-9" siclog@
    // ATTACH ACCEPT, ATTACH COMPLETE
    /*Attach result = 'GPRS only attached'
     * Allocated P-TMSI = P-TMSI-2
     * P-TMSI Signature = P-TMSI-2 signature
     * Routing area identity = RAI-1
     * Periodic RA update timer = 4 minutes
     * T3312 extended value = 6 min
     */
    f_UTRAN34_AttachAccept(utran34_CellA,
                           v_RRC_DataInd.msg.attachRequest.attachType.attachType,
                           tsc_PTMSISig_2,
                           tsc_PTMSI_2,
                           omit,
                           true);
    //@siclog "Step 10" siclog@
    //The SS releases the RRC connection.
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    //Start timer to verify if the time between the attach and the periodic RA updating is 6 minutes
    t_T3312Ext_LowerBound.start;
    t_T3312Ext_UpperBound.start;
    //@siclog "Step 11" siclog@
    alt {
      [] U_TM.receive(car_RRC_ConnReq(utran34_CellA,
                                      tsc_RB0,
                                      cr_RRC_RrcConnReqAny))
        {
          //RRC Connection Request message should not be received within the lower bound
          f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 13");
        }
      [] t_T3312Ext_LowerBound.timeout {
        alt {
          [] U_TM.receive(car_RRC_ConnReq(utran34_CellA,
                                          tsc_RB0,
                                          cr_108_RRC_ConnReq_r8(registration))) -> value v_RLC_TR_DATA_IND
            {
              t_T3312Ext_UpperBound.stop;
              f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 11");
              f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 13");
              v_InitialUE_Id := v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.initialUE_Identity;
              f_UTRAN34_RRC_ConnEst_WithoutRRCConnectionReq(utran34_CellA, v_InitialUE_Id);
            }
          []t_T3312Ext_UpperBound.timeout
            {
              //upper bound timer time-out
              f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 13");
            }
        }
      }
    }
    //@siclog "Step 12" siclog@
    /*Update type = 'Periodic updating'
     * Old P-TMSI signature=P-TMSI-2 signature
     * Old Routing area identity = RAI-1
     */
    U_Dc.receive(car_PS_UplinkDirectTransfer(utran_CellDedicated,
                                             tsc_RB3,
                                             cr_U_RA_UpdReqAny(cr_GMM_UpdateType(?, tsc_I_UpdateTypePeriodic),
                                                               v_RAI_1,
                                                               cr_PTMSI_Signature(tsc_PTMSISig_2) ,
                                                               ?)))
      -> value v_RRC_DataInd;
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 12");
    //@siclog "Step 14" siclog@
    //The SS starts integrity protection
    f_UTRAN34_RRC_Security(utran34_CellA,     // SECURITY MODE COMMAND, SECURITY MODE COMPLETE
                           false,
                           ps_domain);
    //@siclog "Step 15" siclog@
    /*No new mobile identity assigned.
     * P-TMSI not included.
     * Update result = 'RA updated'
     * P-TMSI-3 signature
     * Routing area identity = RAI-1
     */
    U_Dc.send(cas_DataReq_PS(utran_CellDedicated,
                             tsc_RB3,
                             cs_U_RA_UpdAcc3(cs_GMM_UpdateResultRA_Updated,
                                             cs_RAI(omit, cs_LAI(omit, v_PLMN, v_LAC), v_RAC),
                                             cs_PTMSI_Signature(tsc_PTMSISig_3),
                                             omit, //New mobile identity
                                             omit)));
   U_Dc.receive(car_PS_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_RAUpdComplete));

    //@siclog "Step 16" siclog@
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
  } // fl_TC_12_4_3_2a_Body


After:

	function fl_TC_12_4_3_2a_Body() runs on UTRAN_PTC
  {
    var template (value) MobileIdentity v_MobileIdPTMSI_1;
    var RRC_DATA_IND v_RRC_DataInd;
    var U_RLC_TM_IND v_RLC_TR_DATA_IND; //@sic R5-150124 sic@
    var boolean v_NewKeyFlag := true;
    var START_Value v_START_Value_ps := '00000000000000000000'B;
    var PLMN_Identity v_PLMN_Identity;
    var octetstring v_PLMN;
    var octetstring v_LAC;
    var octetstring v_RAC;
    var template (present) RoutingAreaId v_RAI_1;
    var InitialUE_Identity v_InitialUE_Id;
    timer t_T3312Ext_UpperBound := 60.0 * 6.0 * 1.1; // 6 minutes + 10 % tolerance
    timer t_T3312Ext_LowerBound := 60.0 * 6.0 * 0.9; // 6 minutes - 10 % tolerance
    v_PLMN_Identity := f_UTRAN_CellInfo_GetPLMNIdentity(utran34_CellA);
    v_PLMN := f_U_Asn2Nas_PlmnId(v_PLMN_Identity);
    v_LAC := f_UTRAN_CellInfo_GetLAC(utran34_CellA);
    v_RAC := f_UTRAN_CellInfo_GetRAC(utran34_CellA);
    v_RAI_1 :=   cr_RAI(omit, cr_LAI(omit, v_PLMN, v_LAC), v_RAC);
    v_MobileIdPTMSI_1 := cs_MobileIdTMSIorPTMSI(/*'18'O WA#12_4_3_2a*/omit, px_PTMSI_Def);
    //@siclog "Step 2" siclog@
    //The UE is powered up or switched on and initiates an attach (see ICS).
    f_UTRAN_PrepareCellRRC_ConnEst(utran34_CellA, utran34_CellA); //WA#12_4_3_2a
    f_UT_SwitchOnUE(UT, false);
    // For non auto attach UE: trigger the Attach procedure via an AT command.
    if (not(pc_AutomaticAttachSwitchON)) {
      f_Delay(5.0);
      f_UT_TriggerAttach(UT);
    }
    //@siclog "Step 3" siclog@
    //SS checks that the IE "Establishment cause" in the received RRC CONNECTION REQUEST message is set to "Registration".
    f_UTRAN34_RRC_ConnEst(utran34_CellA, delayTolerantAccess /*WA#12_4_3_2 registration*/); 
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 3");
    //@siclog "Step 4" siclog@
    /*Attach type = 'GPRS attach'
     * Mobile identity = P-TMSI-1
     * Old Routing area identity = RAI-1
     * "MS network feature support": 1 (MS supports the extended periodic timer in this domain)
     */
    U_Dc.receive(car_PS_InitDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U34_AttachReq(cr_AttachType(?, tsc_I_AttachTypeGPRSAtt),
                                                                                          v_MobileIdPTMSI_1,
                                                                                          v_RAI_1,
                                                                                          ?)))
      -> value v_RRC_DataInd;
    if (v_RRC_DataInd.msg.attachRequest.msNetworkFeatureSupport.extdPeriodicTimers == '1'B) {
      f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 4");
    }
    else {
      f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 4");
    }
    v_START_Value_ps := v_RRC_DataInd.ttcn_start;
    f_UTRAN_SS_SecurityDownloadStart(ps_domain, v_START_Value_ps);
    //@siclog "Steps 5-6" siclog@
    f_UTRAN_GMM_Authentication();
    //@siclog "Step 7" siclog@
    f_UTRAN34_RRC_Security(utran34_CellA,     // SECURITY MODE COMMAND, SECURITY MODE COMPLETE
                           v_NewKeyFlag,
                           ps_domain);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 6");  //WA#12_4_3_2a
    //@siclog "Steps 8-9" siclog@
    // ATTACH ACCEPT, ATTACH COMPLETE
    /*Attach result = 'GPRS only attached'
     * Allocated P-TMSI = P-TMSI-2
     * P-TMSI Signature = P-TMSI-2 signature
     * Routing area identity = RAI-1
     * Periodic RA update timer = 4 minutes
     * T3312 extended value = 6 min
     */
    f_UTRAN34_AttachAccept(utran34_CellA,
                           v_RRC_DataInd.msg.attachRequest.attachType.attachType,
                           tsc_PTMSI_2 ,/*WA#12_4_3_2a tsc_PTMSISig_2*/
                           tsc_PTMSISig_2 ,/*WA#12_4_3_2a tsc_PTMSI_2*/
                           omit,
                           true);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 9");  //WA#12_4_3_2a
    //@siclog "Step 10" siclog@
    //The SS releases the RRC connection.
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 10");  //WA#12_4_3_2a
    //Start timer to verify if the time between the attach and the periodic RA updating is 6 minutes
    t_T3312Ext_LowerBound.start;
    t_T3312Ext_UpperBound.start;
    //@siclog "Step 11" siclog@
    alt {
      [] U_TM.receive(car_RRC_ConnReq(utran34_CellA,
                                      tsc_RB0,
                                      cr_RRC_RrcConnReqAny))
        {
          //RRC Connection Request message should not be received within the lower bound
          f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 13");
        }
      [] t_T3312Ext_LowerBound.timeout {
        alt {
          [] U_TM.receive(car_RRC_ConnReq(utran34_CellA,
                                          tsc_RB0,
                                          cr_108_RRC_ConnReq_r8(delayTolerantAccess /*WA#12_4_3_2a registration*/))) -> value v_RLC_TR_DATA_IND
            {
              t_T3312Ext_UpperBound.stop;
              f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 11");
              f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 13");
              v_InitialUE_Id := v_RLC_TR_DATA_IND.data.tM_message.uL_CCCH_Message.message_.rrcConnectionRequest.initialUE_Identity;
              f_UTRAN34_RRC_ConnEst_WithoutRRCConnectionReq(utran34_CellA, v_InitialUE_Id);
            }
          []t_T3312Ext_UpperBound.timeout
            {
              //upper bound timer time-out
              f_UTRAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step 13");
            }
        }
      }
    }
    //@siclog "Step 12" siclog@
    /*Update type = 'Periodic updating'
     * Old P-TMSI signature=P-TMSI-2 signature
     * Old Routing area identity = RAI-1
     */
    U_Dc.receive(car_PS_InitDirectTransfer /*WA#12_4_3_2a car_PS_UplinkDirectTransfer*/(utran_CellDedicated,
                                             tsc_RB3,
                                             cr_U_RA_UpdReqAny(cr_GMM_UpdateType(?, tsc_I_UpdateTypePeriodic),
                                                               v_RAI_1,
                                                               cr_PTMSI_Signature(tsc_PTMSISig_2) ,
                                                               ?)))
      -> value v_RRC_DataInd;
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 12");
    //@siclog "Step 14" siclog@
    //The SS starts integrity protection
    f_UTRAN34_RRC_Security(utran34_CellA,     // SECURITY MODE COMMAND, SECURITY MODE COMPLETE
                           false,
                           ps_domain);
    //@siclog "Step 15" siclog@
    /*No new mobile identity assigned.
     * P-TMSI not included.
     * Update result = 'RA updated'
     * P-TMSI-3 signature
     * Routing area identity = RAI-1
     */
    U_Dc.send(cas_DataReq_PS(utran_CellDedicated,
                             tsc_RB3,
                             cs_U_RA_UpdAcc3(cs_GMM_UpdateResultRA_Updated,
                                             cs_RAI(omit, cs_LAI(omit, v_PLMN, v_LAC), v_RAC),
                                             cs_PTMSI_Signature(tsc_PTMSISig_3),
                                             omit, //New mobile identity
                                             omit)));
   /*WA#12_4_3_2a : RAU Complete not expected as per Prose U_Dc.receive(car_PS_UplinkDirectTransfer(utran_CellDedicated, tsc_RB3, cr_U_RAUpdComplete));*/
    //@siclog "Step 16" siclog@
    f_UTRAN34_RRC_ConnRel(utran34_CellA, cell_Dch);
    f_UTRAN_PreliminaryPass(__FILE__, __LINE__, "Step 16");
  } // fl_TC_12_4_3_2a_Body


4.5 TC_12_4_3_2a()
	Record Name
	TC_12_4_3_2a()

	Reason for change
	In the curent TTCN implementation, the guard timer for this testcse in insufficient and needs to be increased.

	Summary of change
	Increased the guard timer from 300s to 600s

	Source of change
	UTRAN_Testsuite.ttcn

	
	


Before:
	testcase TC_12_4_3_2a() runs on MTC_UTRAN system SYSTEM_UTRAN {
    // @purpose
    //   Periodic routing area updating / accepted / per-device timer
    var UTRAN_PTC    v_UTRAN      := null;
    var GERAN_PTC    v_GERAN      := null;
    var EUTRA_PTC    v_EUTRA      := null;
    timer t_GuardTimer := int2float(300); 
    v_UTRAN := UTRAN_PTC.create alive;
    f_MTC_ConnectPTCs_UTRAN(v_UTRAN, v_GERAN, v_EUTRA);
    v_UTRAN.start(f_TC_12_4_3_2a_UTRAN());
    t_GuardTimer.start;
    f_MTC_MainLoop(t_GuardTimer);
  }


After:
	testcase TC_12_4_3_2a() runs on MTC_UTRAN system SYSTEM_UTRAN {
    // @purpose
    //   Periodic routing area updating / accepted / per-device timer
    var UTRAN_PTC    v_UTRAN      := null;
    var GERAN_PTC    v_GERAN      := null;
    var EUTRA_PTC    v_EUTRA      := null;
    timer t_GuardTimer := int2float(600); //WA#12_4_3_2a 300->600
    v_UTRAN := UTRAN_PTC.create alive;
    f_MTC_ConnectPTCs_UTRAN(v_UTRAN, v_GERAN, v_EUTRA);
    v_UTRAN.start(f_TC_12_4_3_2a_UTRAN());
    t_GuardTimer.start;
    f_MTC_MainLoop(t_GuardTimer);
  }


4.6 Cr_RAURequest_Common
	Record Name
	Cr_RAURequest_Common

	Reason for change
	In the curent TTCN implementation, the IE UE_NetworkCap is being checked against template cr_UENetworkCap_Any ('58'O) . This IE is an optional IE, and should be adressed as such. This needs to be changed.

	Summary of change
	Added “if present” for the IE UE_NetworkCap

	Source of change
	NAS_24008Templates_IRAT.ttcn

	
	


Before:
	template (present) ROUTINGAREAUPDATEREQUEST cr_RAURequest_Common(template (present) UpdateType p_UpdateType,
                                                                   template (present) RoutingAreaId p_RAI,
                                                                   template PTMSI_Signature p_PTMSISig,
                                                                   template TMSI_Status p_TMSIStatus,
                                                                   template (present) KeySeq p_KeySeq,
                                                                   template (present) MS_RadioAccessCap_LV  p_MsRadioAccessCap := cr_MS_RadioAccessCapLVAny,
                                                                   template GPRS_Timer        p_ReadyTimer := cr_GPRS_TimerAny ifpresent,
                                                                   template DRXparameter      p_DrxParameter := cr_DRXparameter_Any('27'O) ifpresent,
                                                                   template MobileIdentity    p_Ptmsi := cdr_MobileIdTMSI_tlv ifpresent,
                                                                   template MS_NetworkCap     p_MsNetworkCap := cr_MS_NetworkCap_Any,
                                                                   template PDP_ContextStatus p_PDP_ContextStatus := cr_PDP_ContextStatusAny,
                                                                   template PS_LCS_Capability p_PS_LCS_Capability := cr_PS_LCS_CapabilityAny ifpresent,
                                                                   template MBMS_ContextStatus p_MBMS_ContextStatus := omit,
                                                                   template UE_NetworkCap     p_UeNetworkCapability := cr_UENetworkCap_Any ('58'O),
                                                                   template MobileIdentity    p_AddMobileId := cr_MobileIdAny('1A'O) ifpresent,
                                                                   template RoutingAreaId2    p_AddOldRAI := cr_RoutingAreaId2 ifpresent,
                                                                   template MS_Clsmk2         p_MsClassmark2 := cr_MS_Clsmk2_Any_tlv ifpresent,
                                                                   template MS_Clsmk3         p_MsClassmark3 := cr_MSCLSMK3_Any ifpresent,
                                                                   template CodecList         p_SupportedCodecList := cr_CodecListAny ifpresent,
                                                                   template PTMSI_Type        p_PTMSI_Type := cr_PTMSI_Type_Any ifpresent,
                                                                   template AdditionalUpdateType p_AddUpdType := cr_AdditionalUpdateTypeAny ifpresent,
                                                                   template NwkResourceId p_TmsiBasedNRIContainer := cr_NwkResourceIdAny ifpresent) :=
  { /* @status    APPROVED (IMS_IRAT, LTE_A, LTE_IRAT, POS, SSNITZ, UTRAN) */
    skipIndicator   := '0000'B,
    gMMProtocolDiscriminator := tsc_GMM_PD,
    msgType         := '00001000'B,
    gprsCiphKeySeqNo := cr_CiphKeySeqNum(omit, p_KeySeq),
    updateType      := p_UpdateType,
    oldRAI          := p_RAI,
    msRadioAccessCap := p_MsRadioAccessCap,
    oldPTMSI_Signature := p_PTMSISig,
    readyTimer      := p_ReadyTimer,
    drxParameter    := p_DrxParameter,
    tmsiStatus      := p_TMSIStatus ifpresent,
    ptmsi           := p_Ptmsi,
    msNetworkCap    := p_MsNetworkCap,
    pDP_ContextStatus := p_PDP_ContextStatus,
    pS_LCS_Capability := p_PS_LCS_Capability,
    mBMS_ContextStatus := p_MBMS_ContextStatus,
    ueNetworkCapability := p_UeNetworkCapability,
    addMobileId      := p_AddMobileId,
    addOldRAI        := p_AddOldRAI,
    msClassmark2     := p_MsClassmark2,
    msClassmark3     := p_MsClassmark3,
    supportedCodecList := p_SupportedCodecList,
    voiceDomainPref    := cr_VoiceDomainPref_Any ifpresent, /* @sic R5s110176 Baseline Moving sic@ */
    ptmsiType        := p_PTMSI_Type,  /* cl. 10.5.5.29 O TV 1 IEI=0xE- @sic R5s120178 Baseline Moving, R5-123307 sic@*/
    deviceProperties := cr_DeviceProperties('D'H) ifpresent,  /* cl. 10.5.7.8 O TV 1 IEI=0xD- @sic R5s120178 Baseline Moving sic@*/
    msNetworkFeatureSupport := cr_MS_NetworkFeatureSupport('C'H) ifpresent,   /* cl. 10.5.1.15 O TV 1 IEI=0xC- @sic R5s120178 Baseline Moving sic@*/
    oldLAI           := cr_LocAreaId2_Any ifpresent,  /* cl. 10.5.5.30 O TLV 7 IEI=0x14 @sic R5s120178 Baseline Moving sic@*/
    addUpdType       := p_AddUpdType,  // @sic R5s130195 Baseline Moving sic@
    tmsiBasedNRIContainer := p_TmsiBasedNRIContainer, // @sic R5s130195 Baseline Moving sic@
    t3324Value        := cr_GPRS_Timer2_3('6A'O) ifpresent, // @sic R5s150329 Baseline Moving sic@
    t3312ExtdValue    := cr_GPRS_Timer2_3('39'O) ifpresent // @sic R5s150329 Baseline Moving sic@
  };


After:
	template (present) ROUTINGAREAUPDATEREQUEST cr_RAURequest_Common(template (present) UpdateType p_UpdateType,
                                                                   template (present) RoutingAreaId p_RAI,
                                                                   template PTMSI_Signature p_PTMSISig,
                                                                   template TMSI_Status p_TMSIStatus,
                                                                   template (present) KeySeq p_KeySeq,
                                                                   template (present) MS_RadioAccessCap_LV  p_MsRadioAccessCap := cr_MS_RadioAccessCapLVAny,
                                                                   template GPRS_Timer        p_ReadyTimer := cr_GPRS_TimerAny ifpresent,
                                                                   template DRXparameter      p_DrxParameter := cr_DRXparameter_Any('27'O) ifpresent,
                                                                   template MobileIdentity    p_Ptmsi := cdr_MobileIdTMSI_tlv ifpresent,
                                                                   template MS_NetworkCap     p_MsNetworkCap := cr_MS_NetworkCap_Any,
                                                                   template PDP_ContextStatus p_PDP_ContextStatus := cr_PDP_ContextStatusAny,
                                                                   template PS_LCS_Capability p_PS_LCS_Capability := cr_PS_LCS_CapabilityAny ifpresent,
                                                                   template MBMS_ContextStatus p_MBMS_ContextStatus := omit,
                                                                   template UE_NetworkCap     p_UeNetworkCapability := cr_UENetworkCap_Any ('58'O) ifpresent,/*WA#12_4_3_2a added if present*/
                                                                   template MobileIdentity    p_AddMobileId := cr_MobileIdAny('1A'O) ifpresent,
                                                                   template RoutingAreaId2    p_AddOldRAI := cr_RoutingAreaId2 ifpresent,
                                                                   template MS_Clsmk2         p_MsClassmark2 := cr_MS_Clsmk2_Any_tlv ifpresent,
                                                                   template MS_Clsmk3         p_MsClassmark3 := cr_MSCLSMK3_Any ifpresent,
                                                                   template CodecList         p_SupportedCodecList := cr_CodecListAny ifpresent,
                                                                   template PTMSI_Type        p_PTMSI_Type := cr_PTMSI_Type_Any ifpresent,
                                                                   template AdditionalUpdateType p_AddUpdType := cr_AdditionalUpdateTypeAny ifpresent,
                                                                   template NwkResourceId p_TmsiBasedNRIContainer := cr_NwkResourceIdAny ifpresent) :=
  { /* @status    APPROVED (IMS_IRAT, LTE_A, LTE_IRAT, POS, SSNITZ, UTRAN) */
    skipIndicator   := '0000'B,
    gMMProtocolDiscriminator := tsc_GMM_PD,
    msgType         := '00001000'B,
    gprsCiphKeySeqNo := cr_CiphKeySeqNum(omit, p_KeySeq),
    updateType      := p_UpdateType,
    oldRAI          := p_RAI,
    msRadioAccessCap := p_MsRadioAccessCap,
    oldPTMSI_Signature := p_PTMSISig,
    readyTimer      := p_ReadyTimer,
    drxParameter    := p_DrxParameter,
    tmsiStatus      := p_TMSIStatus ifpresent,
    ptmsi           := p_Ptmsi,
    msNetworkCap    := p_MsNetworkCap,
    pDP_ContextStatus := p_PDP_ContextStatus,
    pS_LCS_Capability := p_PS_LCS_Capability,
    mBMS_ContextStatus := p_MBMS_ContextStatus,
    ueNetworkCapability := p_UeNetworkCapability,
    addMobileId      := p_AddMobileId,
    addOldRAI        := p_AddOldRAI,
    msClassmark2     := p_MsClassmark2,
    msClassmark3     := p_MsClassmark3,
    supportedCodecList := p_SupportedCodecList,
    voiceDomainPref    := cr_VoiceDomainPref_Any ifpresent, /* @sic R5s110176 Baseline Moving sic@ */
    ptmsiType        := p_PTMSI_Type,  /* cl. 10.5.5.29 O TV 1 IEI=0xE- @sic R5s120178 Baseline Moving, R5-123307 sic@*/
    deviceProperties := cr_DeviceProperties('D'H) ifpresent,  /* cl. 10.5.7.8 O TV 1 IEI=0xD- @sic R5s120178 Baseline Moving sic@*/
    msNetworkFeatureSupport := cr_MS_NetworkFeatureSupport('C'H) ifpresent,   /* cl. 10.5.1.15 O TV 1 IEI=0xC- @sic R5s120178 Baseline Moving sic@*/
    oldLAI           := cr_LocAreaId2_Any ifpresent,  /* cl. 10.5.5.30 O TLV 7 IEI=0x14 @sic R5s120178 Baseline Moving sic@*/
    addUpdType       := p_AddUpdType,  // @sic R5s130195 Baseline Moving sic@
    tmsiBasedNRIContainer := p_TmsiBasedNRIContainer, // @sic R5s130195 Baseline Moving sic@
    t3324Value        := cr_GPRS_Timer2_3('6A'O) ifpresent, // @sic R5s150329 Baseline Moving sic@
    t3312ExtdValue    := cr_GPRS_Timer2_3('39'O) ifpresent // @sic R5s150329 Baseline Moving sic@
  };


4.7 Branches executed in test case 12.4.3.2a
The test case implementation was executed with Integrity and ciphering enabled.

5 Execution Log Files

5.1 Intel 7260 device
The Intel 7260 device passed this test case on Rohde & Schwarz CMW500. The documentation below is enclosed as evidence of the successful test case run [1]:

Test Case Execution log file: 
Rohde-Schwarz\tc_12_4_3_2a_Intel_7260.log 
(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)
In the log provided, masking is applied to prevent any disclosure of confidential information.
In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file
6 References
	[1]
	R5s150683
This archive comprises Execution log files, PICS/PIXIT files


